

Электронные таблицы

§ 26. Условные вычисления

§ 27. Обработка больших массивов
данных

§ 28. Численные методы

§ 29. Оптимизация

Электронные таблицы

§ 26. Условные вычисления

Что такое условные вычисления?

Доставка = 20% от стоимости заказа, если > 500р. –
бесплатно.

	A	B	C
1	Заказ	Сумма	Доставка
2	1234	256 р.	51 р.
3	1345	128 р.	26 р.
4	1456	1024 р.	0 р.
5	1565	512 р.	0 р.

```
if B2>500 then  
    C2:=0  
else  
    C2:=B2*0.2;
```

условие

если «да»

если «нет»

```
=IF (B2>500 ; 0 ; B2*0,2)
```

```
=ЕСЛИ (B2>500 ; 0 ; B2*0,2)
```

Символьные данные

Скидка 20% на все товары фирмы «Салют».

	A	B	C	D
1	Код товара	Фирма	Цена	Скидка
2	1234	Салют	3999 р.	800 р.
3	1345	Звезда	2799 р.	
4	1456	Гамбит	6290 р.	
5	1565	Салют	3750 р.	750 р.
6	1576	Гамбит	1234 р.	

`=IF (B2="Салют" ; C2*20% ; "")`

`=ЕСЛИ (B2="Салют" ; C2*20% ; "")`

`C2*0,2`

Пример

Работник получает премию, составляющую 10% от его зарплаты, только тогда, когда на него не было жалоб.

	A	B	C	D
1	<i>Фамилия</i>	<i>Зарплата</i>	<i>Жалобы</i>	<i>Премия</i>
2	Иванов	12 000 р.	0	1 200 р.
3	Петров	14 000 р.	2	0 р.



=IF (C2=0 ; B2*10% ; 0)

=ЕСЛИ (C2=0 ; B2*10% ; 0)

Вложенные вызовы ЕСЛИ

Доставка = если > 500 р. – бесплатно; если более 200р. (но ≤ 500 р.), то 10% от стоимости заказа, если ≤ 200 р., то 20% от стоимости заказа.

	A	B	C
1	Заказ	Сумма	Доставка
2	1234	256 р.	26 р.
3	1345	128 р.	26 р.
4	1456	1024 р.	0 р.

```

if B2>500 then
  C2:=0
else
  if B2>200 then
    C2:=B2*0.1
  else C2:=B2*0.2;
  
```

 Сколько вариантов?

=IF (B2>500 ; 0 ; IF (B2>200 ; B2*10% ; B2*20%))

=ЕСЛИ (B2>500 ; 0 ; ЕСЛИ (B2>200 ; B2*10% ; B2*20%))

Сложные условия – «И»

Доставка бесплатна, если номер заказа < 1500 и сумма > 500р., иначе 20% от стоимости заказа.

	A	B	C
1	Заказ	Сумма	Доставка
2	1234	256 р.	52 р.
3	2345	1580 р.	316 р.
4	1456	1024 р.	0 р.

условие 1

условие 2

```
if (A2<1500) and  
    (B2>500) then  
    C2:=0  
else  
    C2:=B2*0.2;
```

```
=IF (AND (A2<1500 ; B2>500) ; 0 ; B2*20%)
```

```
=ЕСЛИ (И (A2<1500 ; B2>500) ; 0 ; B2*20%)
```

Сложные условия – «ИЛИ»

Разрешается ехать со скоростью от 40 км/ч до 110 км/ч.
При другой скорости – штраф 500р.

	А	В	С
1	Номер	Скорость	Штраф
2	A134AA	150	500 р.
3	B235BB	80	
4	A157AB	90	
5	A198CX	30	500 р.
6	K754MM	180	500 р.

условие 1

условие 2

```
=IF ( OR (B2<40 ; B2>110) ; 500 ; "" )
```

```
=ЕСЛИ ( ИЛИ (B2<40 ; B2>110) ; 500 ; "" )
```


Больше двух условий

Проход на III тур: набрать по сумме двух первых туров не менее 180 баллов или получить 100 баллов хотя бы в одном туре.

	A	B	C	D
1	Участник	I тур	II тур	III тур
2	Иванов И.И.	100	70	+
3	Петров П.П.	80	75	
4	Сидоров С.С.	65	100	+
5	Куницын К.К.	95	90	+
6	Васильев В.В.	80	90	



```
=IF (OR (B2+C2>=180 ; B2=100 ; C2=100) ; "+" ; "")
```


```
=ЕСЛИ (OR (B2+C2>=180 ; B2=100 ; C2=100) ; "+" ; "")
```

Электронные таблицы


§ 27. Обработка больших массивов данных

Выделение больших диапазонов

Протаскивание мыши



Два щелчка в противоположных углах:



Выделение больших диапазонов

До первой пустой ячейки в нужном направлении:

	A	B	C	D
1				
2		1	2	
3		11	22	
4		111	222	
5				

Ctrl+ Shift+↓

В поле *Имя* над таблицей:

ввести адрес
диапазона

	A	B	C	D
1				
2				

А5:В1000

Вспомогательные столбцы

Не используя сортировку, найти количество спортсменов 2004 года рождения.

	А	В	С	Д
1	Участник	Год рождения	Вес, кг	
2	Иванов И.И.	2004	56	1
3	Петров П.П.	2003	62	0
4	Сидоров С.	2004	58	1
			...	



`=IF (B2=2004 ; 1 ; 0)`

`=ЕСЛИ (B2=2004 ; 1 ; 0)`



Ответ?

`=SUM (D2 : D1001)`

`=СУММА (D2 : D1001)`

Вспомогательные столбцы

Не используя сортировку, найти средний вес спортсменов 2004 года рождения.

	А	В	С	Д
1	Участник	Год рождения	Вес, кг	
2	Иванов И.И.	2004	56	56
3	Петров П.П.	2003	62	
	Сидоров С.	2004	58	58

`=IF (B2=2004 ; C2 ; " ")`

`=ЕСЛИ (B2=2004 ; C2 ; " ")`



Почему не 0?



Ответ?



`=AVERAGE (D2 : D1001)`

`=СРЗНАЧ (D2 : D1001)`

Функции COUNT и COUNTIF

Количество числовых ячеек в диапазоне:

```
=COUNT (D2:D1001)
```

```
=СЧЁТ (D2:D1001)
```

Количество ячеек с заданным значением:

```
=COUNTIF (D2:D1001 ; 2004)
```

```
=СЧЁТЕСЛИ (D2:D1001 ; 2004)
```

или так

```
=COUNTIF (D2:D1001 ; "=2004")
```

```
=СЧЁТЕСЛИ (D2:D1001 ; "=2004")
```

текстовое значение:

```
=COUNTIF (D2:D1001 ; "яблоко")
```

```
=СЧЁТЕСЛИ (D2:D1001 ; "яблоко")
```

Функции COUNTIF и COUNTIFS

Количество ячеек, удовлетворяющих условию:

```
=COUNTIF (D2:D1001 ; ">2004" )
```

```
=СЧЁТЕСЛИ (D2:D1001 ; ">2004" )
```



Нельзя использовать сложное условие!

```
=COUNTIFS (D2:D101 ; ">2004" ;  
             E2:E101 ; "<59" )
```

```
=СЧЁТЕСЛИМН (D2:D101 ; ">2004" ;  
             E2:E101 ; "<59" )
```


Пример



Определить, какую долю (в процентах) от общего количества составляют спортсмены 2004 года рождения:

```
=COUNTIF (D2:D1001 ; 2004) / COUNT (D2:D1001)
```

```
=СЧЁТЕСЛИ (D2:D1001 ; 2004) / СЧЁТ (D2:D1001)
```



И установить процентный формат!



настройка количества знаков

Функция SUMIF

Найти общий вес спортсменов 2004 года рождения:

	А	В	С
1	Участник	Год рождения	Вес, кг
2	Иванов И.И.	2004	56
3	Петров П.П.	2003	62
	Сидоров С.	2004	58

диапазон проверки условия

условие

диапазон суммирования

```
=SUMIF (B2 : B1001 ; 2004 ; C2 : C1001)
```

```
=СУМЕСЛИ (B2 : B1001 ; 2004 ; C2 : C1001)
```

Функция AVERAGEIF

Найти средний вес спортсменов 2004 года рождения:

	А	В	С
1	Участник	Год рождения	Вес, кг
2	Иванов И.И.	2004	56
3	Петров П.П.	2003	62
4	Сидоров С.	2004	58

диапазон проверки условия


условие

диапазон для вычисления среднего

```
=AVERAGEIF (B2 : B1001 ; 2004 ; C2 : C1001 )
```

```
=СРЗНАЧЕСЛИ (B2 : B1001 ; 2004 ; C2 : C1001 )
```

Работа с листами



The screenshot shows a spreadsheet application interface. On the left, a grid of cells is visible with row numbers 36, 37, 38, and 39. A yellow callout box with the text "ПКМ" (Right Mouse Button) points to a cell in row 37. A context menu is open over the spreadsheet, listing several options: "Вставить..." (Paste...), "Удалить" (Delete), "Переименовать" (Rename), "Переместить/скопировать..." (Move/Copy...), "Защитить лист..." (Protect Sheet...), and "Выделить все листы" (Select All Sheets). The spreadsheet tabs at the bottom show "Январь" (January) and "Февраль" (February).

Обращение к данным другого листа:

OpenOffice Calc:

=Январь . B2+Февраль . B2+Март . B2

=SUM (' К оплате ' . B2 : C4)

Microsoft Excel:

=Январь ! B2+Февраль ! B2+Март ! B2

=СУММ (' К оплате ' ! B2 : C4)

Электронные таблицы

§ 28. Численные методы

Методы решения уравнений

Точные (аналитические) методы:

$$ax + b = 1, \quad a \neq 0 \quad \Rightarrow \quad x = \frac{1-b}{a}$$



Можем понять зависимость x от a и b !

$$a \cdot x = b \cdot \cos x$$



Как решать?

Численное решение – это решение задачи для конкретных исходных данных.

$$a = 1, b = 1, \quad \Rightarrow \quad x = \cos x$$

Численный метод – это метод, который применяется для поиска численного решения.

Численные методы



Численные методы часто дают неточное (приближенное решение)!

Приближённый метод – это метод, который позволяет найти решение задачи с некоторой (допустимой) ошибкой (погрешностью).

Погрешность — отклонение значения величины, полученного в результате измерений или вычислений, от её истинного (действительного) значения.




$$x \neq 0,517$$

$$x \approx 0,517$$

Пример

$x = \cos x$  Как решать?

Графический метод:



Можно поручить такой поиск компьютеру!




Можно ли получить точное решение?

Как работают численные методы?

Сжатие отрезка:

- 1) выбрать начальный отрезок $[a_0, b_0]$ (одно решение!)
- 2) уточнить решение с помощью некоторого алгоритма: $\Rightarrow [a, b]$
- 3) повторять шаг 2, пока длина отрезка $[a, b]$ не станет достаточно мала



Что лучше выбрать в качестве решения?



Как оценить ошибку?

$$|x - x^*| \leq \frac{b - a}{2}$$

Завершение работы:

$$b - a \leq 2\varepsilon$$

допустимая
ошибка

Как работают численные методы?

По одной точке:

- 1) выбрать начальное приближение x_0

Начальное приближение – это начальное значение неизвестной величины, которое уточняется с помощью приближённого метода.

- 2) уточнить решение с помощью некоторого алгоритма:

$$\Rightarrow x_1 \Rightarrow x_2 \Rightarrow x_3 \Rightarrow \dots$$

- 3) повторять шаг 2, пока два последовательных приближения не будут достаточно мало

Завершение работы: $|x_i - x_{i-1}| \leq \varepsilon$

Численные методы: «за» и «против»



- дают практическое решение задачи



- неточное решение $x \approx 0,517$
- неясно, как зависит от исходных данных (параметров)
- объём вычислений может быть велик
- не всегда легко оценить ошибку

Пример решения уравнения

Найти все решения на интервале $[-2; 2]$: $x^2 = \cos x$

$$f_1(x) = x^2 \quad f_2(x) = \cos x$$



Сколько решений?

→ строим графики

Заполнение таблицы

	A	B	C
1	x	$f_1(x)$	$f_2(x)$
2	-2		
3	-1,75		
4			
5			
6			
7			
8			



	A	B	C
1	x	$f_1(x)$	$f_2(x)$
2	-2	=A2^2	=COS(A2)
3	-1,75		
4	-1,5		
5	-1,25		
6	-1		
7	-0,75		
8	-0,5		

Пример решения уравнения

Построение графиков

OpenOffice Calc:

Диаграмма X-Y


Microsoft Excel:

Точечная

- всего 2 решения
- начальные приближения:

$$x_1 = -1$$

$$x_2 = 1$$



Пример решения уравнения

Подготовка данных

начальное
приближение

	E	F	G	H
1	x	$f_1(x)$	$f_2(x)$	$f_1(x) - f_2(x)$
2	-1	=E2^2	=COS(E2)	=F2-G2
3				



Зачем нужна разность функций?



Если x – решение уравнения $f_1(x) = f_2(x)$, то
 $f_1(x) - f_2(x) = 0!$

Пример решения уравнения

Подбор параметра

изменяемая
ячейка

	Е	Ф	Г	Н
	x	$f_1(x)$	$f_2(x)$	$f_1(x) - f_2(x)$
2	-1	=E2^2	=COS(E2)	=F2 - G2
3				

целевая
ячейка

OpenOffice Calc:

Сервис – Подбор параметра

Microsoft Excel:

Данные – Анализ «что-если» – Подбор параметра



Как найти второе
решение?



Электронные таблицы

§ 29. Оптимизация

Что такое оптимизация?

Оптимизация – это поиск наилучшего (*оптимального*) решения задачи в заданных условиях.

1) **Цель**: выбрать неизвестный x , так чтобы

$$f(x) \rightarrow \min$$

или $f(x) \rightarrow \max$

целевая функция

$$- f(x) \rightarrow \min$$


2) **Ограничения**

задача
оптимизации



Почему неправильно «самый оптимальный»?

Что такое минимум?




- обычно нужно найти **глобальный** минимум
- большинство численных методов находят только **локальный** минимум



Результат локальной оптимизации зависит от начального приближения!

Оптимальный раскрой листа



Цель: $V(x) \rightarrow \max$ $V(x) = x \cdot (1 - 2x)^2 \rightarrow \max$


? Какие ограничения?

Ограничения: $0 < x < 0,5$

? Какой результат ожидаете (по интуиции)?

Оптимальный раскрой листа

В табличном процессоре:



$$V(x) = x \cdot (1 - 2x)^2 \rightarrow \max$$

начальное
приближение $\approx 0,2$



Какая формула в F2?

	E	F
1	x	Объём
2	0,2	0,072

Оптимизация в табличном процессоре

Задача оптимизации: найти максимум (или минимум) целевой функции в ячейке ..., изменяя значения ячеек ... при ограничениях

OpenOffice.org Calc:

Сервис – Поиск решения

Microsoft Excel:

надстройка Данные – Поиск решения

Оптимизация в табличном процессоре

OpenOffice.org Calc:

Решатель

Целевая ячейка:

Оптимизация результата: Максимум
 Минимум
 Значение

Путем изменения ячеек:

Ограничительные условия

Ссылка на ячейку	Операция	Значение
<input type="text" value="\$E\$2"/>	<input type="text" value=">="/>	<input type="text" value="0"/>
<input type="text" value="\$E\$2"/>	<input type="text" value="<="/>	<input type="text" value="0,5"/>
<input type="text"/>	<input type="text" value="<="/>	<input type="text"/>
<input type="text"/>	<input type="text" value="<="/>	<input type="text"/>

Параметры... Справка Закрыть

целевая
ячейка

изменяемые
ячейки:

E2

D2:D6

D2:D6; C5:C8

ограничения

A1 <= 20


A1 = целое

B2:B8 >= 5

B2:B4 >= C2:C4

Оптимизация в табличном процессоре

Excel:



Конец фильма

ПОЛЯКОВ Константин Юрьевич

д.т.н., учитель информатики

ГБОУ СОШ № 163, г. Санкт-Петербург

kpolyakov@mail.ru

ЕРЕМИН Евгений Александрович

к.ф.-м.н., доцент кафедры мультимедийной

дидактики и ИТО ПГГПУ, г. Пермь

eremin@pspu.ac.ru

Источники иллюстраций

1. иллюстрации художников издательства «Бином»
2. авторские материалы